Pensieve header: Nilpotent Gaussian integration.

Initialization

```
In[•]:= CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
CCF[ℰ_] := Factor[ℰ];
CF[ω_. ℰ_𝔼] := CF[ω] CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p)__, ∞] ⋃ {x, p, ϵ}, ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ CCF[c] (Times @@ vs^ps)] ];
```

Integration

Using Picard Iteration!

```
In[•]:= 𝔼 /: 𝔼[A_] 𝔼[B_] := 𝔼[A + B]
```

```
In[•]:= $π = Identity;
```

What must a Feynman Ring F have? (Over some set of labels S with elements x, y, ...)

* A vector space over $\mathbb{Q}$.

* Has a symmetric linear $Z \longmapsto \partial_{x,y} Z$ and a symmetric bilinear $(Z_1, Z_2) \longmapsto (\partial_x Z_1)(\partial_y Z_2)$ that satisfy the axioms of (roughly) a connected circuit algebra.

* Has $q_{x,y} : F \to \mathbb{Q}$ in some sense dual to some $\theta_{x,y} \in F$.

* Has $\mathrm{Ev}_{vs\to 0} : F \to F$.

Further axioms must be worked out.

Goals.

* Define $\int$.

* Prove a Fubini theorem.

* Prove a theorem about the injectivity of the Laplace transform.

In[∘]:=
```
Unprotect[Integrate];
Integrate::sing = "How dare you ask me to integrate a singular Gaussian!";
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, L0, Q, Δ, G, Z, e, λ, a, b},
   n = Length@vs; L0 = L /. ϵ → 0;
   Q = Table[(-∂_{vs⟦a⟧,vs⟦b⟧} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = CF@Det[Q]) == 0, Message[Integrate::sing]; Return[]];
   Z = CF@$π[L + vs.Q.vs / 2]; G = Inverse[Q] / 2;
   While[
    e = CF@$π[(∂_λ Z) - Sum[G⟦a, b⟧ ((∂_{vs⟦a⟧,vs⟦b⟧} Z) + (∂_{vs⟦a⟧} Z) (∂_{vs⟦b⟧} Z)),
        {a, n}, {b, n}]];
    0 =!= e, Z -= ∫_0^λ e ⅆλ
   ];
   PowerExpand@Factor[ω (Δ (2 π)^n)^(-1/2)] 𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]
  ];
Protect[Integrate];
```

In[∘]:= $\int \mathbb{E}\left[i\lambda x_1^2/2\right] dd\{x_1\}$

Out[∘]=

$$\frac{(-1)^{1/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-i\lambda x_1^2/2\right] dd\{x_1\}$

Out[∘]=

$$-\frac{(-1)^{3/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{i}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c\end{pmatrix}.\{x_1, x_2\}\right] dd\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{2\sqrt{b^2 - a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda x_1^2/2\right] dd\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-x_1^2/2+\xi\, x_1\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{\xi^2}{2}\right]}{\sqrt{2\pi}}$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}+\{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2-2\,b\,\xi_1\,\xi_2+a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:= $I1 = \int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}+\{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[-\frac{(-b^2+a\,c)\,x_2^2}{2\,a}+\frac{\xi_1^2}{2\,a}+\frac{x_2\,(-b\,\xi_1+a\,\xi_2)}{a}\right]}{\sqrt{a}\;\sqrt{2\pi}}$$

In[∘]:= $\int I1\, d\{x_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2-2\,b\,\xi_1\,\xi_2+a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[\xi\,x+\eta\,y+z\,(x-y)+x^2\right] d\{x, z\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}[y\,(y+\eta+\xi)]}{2\pi}$$

Integration of ϵ-Series

In[∘]:= $\text{Block}\left[\left\{\$\pi = \text{Normal}\left[\#+O[\epsilon]^7\right]\&\right\},\ \int \mathbb{E}\left[-x^2/2+\epsilon\,x^3/6\right] d\{x\}\right]$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{5\,\epsilon^2}{24}+\frac{5\,\epsilon^4}{16}+\frac{1105\,\epsilon^6}{1152}\right]}{\sqrt{2\pi}}$$

In[ ]:=
```
Block[{$π = Normal[# + O[ϵ]^7] &},
 ∫𝔼[-ϕ^2/2 + ϵ ϕ^4/24] ⅆ{ϕ}]
```

Out[ ]=

$$\frac{\mathbb{E}\left[\frac{\epsilon}{8}+\frac{\epsilon^2}{12}+\frac{11\,\epsilon^3}{96}+\frac{17\,\epsilon^4}{72}+\frac{619\,\epsilon^5}{960}+\frac{709\,\epsilon^6}{324}\right]}{\sqrt{2\,\pi}}$$

In[ ]:=
```
Block[{$π = Normal[# + O[ϵ]^5] &},
 ∫𝔼[p x + ϵ p^2 x] ⅆ{p, x}]
```

Out[ ]=

$$-\frac{i\,\mathbb{E}[0]}{2\,\pi}$$

In[ ]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
     mon ↦ And[
       Exponent[mon, ϵ] ≤ 2,
       Exponent[mon, x] == Exponent[mon, p]
      ]
    ] &},
 ∫𝔼[p x + a x^2 p + ϵ b x^3 p^3] ⅆ{p, x}]
```

Out[ ]=

$$-\frac{i\,\mathbb{E}\left[-6\,b\,\epsilon+342\,b^2\,\epsilon^2\right]}{2\,\pi}$$

In[ ]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
     mon ↦ And[
       Exponent[mon, ϵ] < 4,
       Exponent[mon, x] - Exponent[mon, p] ≤ 3
      ]
    ] &},
 ∫𝔼[p x + a x^2 p + ϵ b p^2 x] ⅆ{p, x}]
```

Out[ ]=

$$-\frac{i\,\mathbb{E}\left[-6\,a\,b\,\epsilon+162\,a^2\,b^2\,\epsilon^2-9072\,a^3\,b^3\,\epsilon^3\right]}{2\,\pi}$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
      mon ↦ And[
        Exponent[mon, ϵ] < 4,
        Exponent[mon, x] - Exponent[mon, p] ≤ 3 - Exponent[mon, ϵ]
       ]
     ] &},
  ∫𝔼[p x + a x^2 p + ϵ b p^2 x] ⅆ{p, x}]
```

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-6\,a\,b\,\epsilon+162\,a^2\,b^2\,\epsilon^2-9072\,a^3\,b^3\,\epsilon^3\right]}{2\,\pi}$$

In[∘]:=
```
MatrixForm@Table[
  ∫𝔼[x_1 p_2 + x_2 p_3 + x_3 p_1 + ξ_i x_i + π_j p_j] ⅆ{x_1, x_2, x_3, p_1, p_2, p_3},
  {i, 3}, {j, 3}]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[-\pi_2\,\xi_1]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} \\ -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[-\pi_3\,\xi_2]}{8\,\pi^3} \\ -\frac{i\,\mathbb{E}[-\pi_1\,\xi_3]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\,\pi^3} \end{pmatrix}$$